Writing a Simple REST Web Service in PureScript - Part 2

- Posted on October 1, 2017 by Abhinav Sarkar
- Tags: purescript, REST, programming, nilenso

② 21 minute read

To recap, in the first part of this two-part tutorial, we built a simple JSON REST web service in PureScript to create, update, get, list and delete users, backed by a Postgres database. In this part we'll work on the rest of the requirements:

- 1. validation of API requests.
- 2. reading the server and database configs from environment variables.
- 3. logging HTTP requests and debugging info.

Contents

- 1. Bugs!
- 2. Validation
- 3. Configuration
- 4. Logging
- 5. Conclusion

But first,

Bugs!

What happens if we hit a URL on our server which does not exist? Let's fire up the server and test it:

```
1  $ pulp --watch run

1  $ http GET http://localhost:4000/v1/random
2  HTTP/1.1  404 Not Found
3  Connection: keep-alive
4  Content-Length: 148
5  Content-Security-Policy: default-src 'self'
6  Content-Type: text/html; charset=utf-8
7  Date: Sat, 30 Sep 2017 08:23:20 GMT
8  X-Content-Type-Options: nosniff
9  X-Powered-By: Express
10
11  <!DOCTYPE html>
12  <html lang="en">
13  <head>
14  <meta charset="utf-8">
15  <title>Error</title>
```

```
16 </head>
17 <body>
18 cpre>Cannot GET /v1/random
19 </body>
20 </html>
```

We get back a default HTML response with a 404 status from Express. Since we are writing a JSON API, we should return a JSON response in this case too. We add the following code in the src/SimpleService/Server.purs file to add a catch-all route and send a 404 status with a JSON error message:

```
1 -- previous code
2 import Data.Either (fromRight)
3 import Data.String.Regex (Regex, regex) as Re
4 import Data.String.Regex.Flags (noFlags) as Re
5 import Node.Express.App (App, all, delete, get, http, listenHttp, post, useExternal)
6 import Node.Express.Response (sendJson, setStatus)
  import Partial.Unsafe (unsafePartial)
  -- previous code
9
10 allRoutePattern :: Re.Regex
11 allRoutePattern = unsafePartial $ fromRight $ Re.regex "/.*" Re.noFlags
12
13 app :: forall eff. PG.Pool -> App (postgreSQL :: PG.POSTGRESQL | eff)
  app pool = do
    useExternal jsonBodyParser
15
16
    17
18
    delete "/v1/user/:id" $ deleteUser pool
19
    patch "/v1/user/:id" $ updateUser pool
20
    21
22
    all allRoutePattern do
23
24
      setStatus 404
     sendJson {error: "Route not found"}
25
26
    where
27
      patch = http (CustomMethod "patch")
```

allRoutePattern matches all routes because it uses a "/.*" regular expression. We place it as the last route to match all the otherwise unrouted requests. Let's see what is the result:

```
1  $ http GET http://localhost:4000/v1/random
2  HTTP/1.1  404 Not Found
3  Connection: keep-alive
4  Content-Length: 27
5  Content-Type: application/json; charset=utf-8
6  Date: Sat, 30 Sep 2017 08:46:46 GMT
7  ETag: W/"1b-772e0u4nrE48ogbR0KmKfSvrHUE"
8  X-Powered-By: Express
```

```
9
10 {
11     "error": "Route not found"
12 }
```

Now we get a nicely formatted JSON response.

Another scenario is when our application throws some uncaught error. To simulate this, we shut down our postgres database and hit the server for listing users:

```
1 $ http GET http://localhost:4000/v1/users
2 HTTP/1.1 500 Internal Server Error
3 Connection: keep-alive
4 Content-Length: 372
5 Content-Security-Policy: default-src 'self'
6 Content-Type: text/html; charset=utf-8
7 Date: Sat, 30 Sep 2017 08:53:40 GMT
8 X-Content-Type-Options: nosniff
9 X-Powered-By: Express
10
11 <!DOCTYPE html>
12 <html lang="en">
13 <head>
14 <meta charset="utf-8">
15 <title>Error</title>
16 </head>
17 <body>
18 Frror: connect ECONNREFUSED 127.0.0.1:5432<br> &nbsp; &nbsp; at
   Object._errnoException (util.js:1026:11)<br/>
or> Object._errnoException(util.js:1026:11)<br/>
ort
   (util.js:1049:20)<br> &nbsp; &nbsp; at TCPConnectWrap.afterConnect [as oncomplete]
   (net.js:1174:14)
19 </body>
20 </html>
```

We get another default HTML response from Express with a 500 status. Again, in this case we'd like to return a JSON response. We add the following code to the src/SimpleService/Server.purs file:

```
1 -- previous code
import Control.Monad.Eff.Exception (message)
import Node.Express.App (App, all, delete, get, http, listenHttp, post, useExternal,
   useOnError)
  -- previous code
  app :: forall eff. PG.Pool -> App (postgreSQL :: PG.POSTGRESQL | eff)
  app pool = do
    -- previous code
8
    useOnError \err -> do
9
      setStatus 500
     sendJson {error: message err}
11
12
    where
```

```
patch = http (CustomMethod "patch")
```

We add the useOnError handler which comes with purescript-express to return the error message as a JSON response. Back on the command-line:

```
1  $ http GET http://localhost:4000/v1/users
2  HTTP/1.1 500 Internal Server Error
3  Connection: keep-alive
4  Content-Length: 47
5  Content-Type: application/json; charset=utf-8
6  Date: Sat, 30 Sep 2017 09:01:37 GMT
7  ETag: W/"2f-cJuIW6961YCpo9TWDSZ9VWHLGHE"
8  X-Powered-By: Express
9
10  {
11    "error": "connect ECONNREFUSED 127.0.0.1:5432"
12 }
```

It works! Bugs are fixed now. We proceed to add next features.

Validation

Let's recall the code to update a user from the src/SimpleService/Handler.purs file:

```
updateUser :: forall eff. PG.Pool -> Handler (postgreSQL :: PG.POSTGRESQL | eff)
   updateUser pool = getRouteParam "id" >>= case _ of
     Nothing -> respond 422 { error: "User ID is required" }
     Just sUserId -> case fromString sUserId of
4
       Nothing -> respond 422 { error: "User ID must be positive: " <> sUserId }
5
       Just userId -> getBody >>= case _ of
         Left errs -> respond 422 { error: intercalate ", " $ map renderForeignError errs}
         Right (UserPatch userPatch) -> case unNullOrUndefined userPatch.name of
           Nothing -> respondNoContent 204
           Just userName -> if userName == ""
10
             then respond 422 { error: "User name must not be empty" }
11
12
             else do
               savedUser <- liftAff $ PG.withConnection pool \conn -> PG.withTransaction
   conn do
                 P.findUser conn userId >>= case of
14
                   Nothing -> pure Nothing
15
                   Just (User user) -> do
16
                     let user' = User (user { name = userName })
18
                     P.updateUser conn user'
                     pure $ Just user'
19
20
               case savedUser of
                 Nothing -> respond 404 { error: "User not found with id: " <> sUserId }
21
22
                 Just user -> respond 200 (encode user)
```

As we can see, the actual request handling logic is obfuscated by the request validation logic for the user id and the user name patch parameters. We also notice that we are using three constructs for validation here:

Maybe, Either and if-then-else. However, we can use just Either to subsume all these cases as it can "carry" a failure as well as a success case. Either also comes with a nice monad transformer ExceptT which provides the do syntax for failure propagation. So we choose ExceptT as the base construct for our validation framework and write functions to upgrade Maybe and if-then-else to it. We add the following code to the src/SimpleService/Validation.purs file:

```
module SimpleService.Validation
     (module MoreExports, module SimpleService.Validation) where
2
3
   import Prelude
  import Control.Monad.Except (ExceptT, except, runExceptT)
7 import Data.Either (Either(..))
8 import Data.Maybe (Maybe(..))
  import Node.Express.Handler (HandlerM, Handler)
10 import Node.Express.Response (sendJson, setStatus)
11 import Node.Express.Types (EXPRESS)
import Control.Monad.Except (except) as MoreExports
13
14 type Validation eff a = ExceptT String (HandlerM (express :: EXPRESS | eff)) a
15
16 exceptMaybe :: forall e m a. Applicative m => e -> Maybe a -> ExceptT e m a
  exceptMaybe e a = except $ case a of
17
     Just x -> Right x
18
19
     Nothing -> Left e
20
21 exceptCond :: forall e m a. Applicative m => e -> (a -> Boolean) -> a -> ExceptT e m a
   exceptCond e cond a = except $ if cond a then Right a else Left e
24 withValidation :: forall eff a. Validation eff a -> (a -> Handler eff) -> Handler eff
25 withValidation action handler = runExceptT action >>= case _ of
    Left err -> do
26
27
       setStatus 422
       sendJson {error: err}
28
     Right x -> handler x
```

We re-export except from the Control. Monad. Except module. We also add a withValidation function which runs an ExceptT based validation and either returns an error response with a 422 status in case of a failed validation or runs the given action with the valid value in case of a successful validation.

Using these functions, we now write updateUser in the src/SimpleService/Handler.purs file as:

```
1  -- previous code
2  import Control.Monad.Trans.Class (lift)
3  import Data.Bifunctor (lmap)
4  import Data.Foreign (ForeignError, renderForeignError)
5  import Data.List.NonEmpty (toList)
6  import Data.List.Types (NonEmptyList)
7  import Data.Tuple (Tuple(..))
8  import SimpleService.Validation as V
```

```
-- previous code
10
   renderForeignErrors :: forall a. Either (NonEmptyList ForeignError) a -> Either String a
   renderForeignErrors = lmap (toList >>> map renderForeignError >>> intercalate ", ")
13
   updateUser :: forall eff. PG.Pool -> Handler (postgreSQL :: PG.POSTGRESQL | eff)
14
   updateUser pool = V.withValidation (Tuple <$> getUserId <*> getUserPatch)
                                       \(Tuple userId (UserPatch userPatch)) ->
16
       case unNullOrUndefined userPatch.name of
17
18
         Nothing -> respondNoContent 204
19
         Just uName -> V.withValidation (getUserName uName) \userName -> do
           savedUser <- liftAff $ PG.withConnection pool \conn -> PG.withTransaction conn
20
   do
21
             P.findUser conn userId >>= case _ of
               Nothing -> pure Nothing
22
23
               Just (User user) -> do
                 let user' = User (user { name = userName })
24
                 P.updateUser conn user'
25
                 pure $ Just user'
26
27
           case savedUser of
             Nothing -> respond 404 { error: "User not found with id: " <> show userId }
28
             Just user -> respond 200 (encode user)
29
30
     where
       getUserId = lift (getRouteParam "id")
31
32
         >>= V.exceptMaybe "User ID is required"
         >>= fromString >>> V.exceptMaybe "User ID must be positive"
33
34
       getUserPatch = lift getBody >>= V.except <<< renderForeignErrors</pre>
35
36
37
       getUserName = V.exceptCond "User name must not be empty" (_ == "")
```

The validation logic has been extracted out in separate functions now which are composed using Applicative. The validation steps are composed using the ExceptT monad. We are now free to express the core logic of the function clearly. We rewrite the src/SimpleService/Handler.purs file using the validations:

```
module SimpleService.Handler where

import Prelude

import Control.Monad.Aff.Class (liftAff)

import Control.Monad.Trans.Class (lift)

import Data.Bifunctor (lmap)

import Data.Either (Either)

import Data.Foreign (ForeignError, renderForeignError)

import Data.Foreign.Class (encode)

import Data.Foreign.NullOrUndefined (unNullOrUndefined)

import Data.Int (fromString)

import Data.List.NonEmpty (toList)

import Data.List.Types (NonEmptyList)
```

```
16 import Data.Maybe (Maybe(..))
17 import Data.Tuple (Tuple(..))
18 import Database.PostgreSQL as PG
19 import Node.Express.Handler (Handler)
20 import Node.Express.Request (getBody, getRouteParam)
21 import Node.Express.Response (end, sendJson, setStatus)
22 import SimpleService.Persistence as P
23 import SimpleService. Validation as V
24 import SimpleService.Types
25
26 getUser :: forall eff. PG.Pool -> Handler (postgreSQL :: PG.POSTGRESQL | eff)
   getUser pool = V.withValidation getUserId \userId ->
     liftAff (PG.withConnection pool $ flip P.findUser userId) >>= case _ of
       Nothing -> respond 404 { error: "User not found with id: " <> show userId }
29
       Just user -> respond 200 (encode user)
30
31
32 deleteUser :: forall eff. PG.Pool -> Handler (postgreSQL :: PG.POSTGRESQL | eff)
33 deleteUser pool = V.withValidation getUserId \userId -> do
     found <- liftAff $ PG.withConnection pool \conn -> PG.withTransaction conn do
34
       P.findUser conn userId >>= case _ of
35
36
         Nothing -> pure false
         Just _ -> do
37
           P.deleteUser conn userId
38
39
           pure true
40
     if found
       then respondNoContent 204
41
       else respond 404 { error: "User not found with id: " <> show userId }
42
44 createUser :: forall eff. PG.Pool -> Handler (postgreSQL :: PG.POSTGRESQL | eff)
45 createUser pool = V.withValidation getUser \usera(User ) -> do
     liftAff (PG.withConnection pool $ flip P.insertUser user)
46
     respondNoContent 201
47
     where
48
49
       getUser = lift getBody
50
         >>= V.except <<< renderForeignErrors
         >>= V.exceptCond "User ID must be positive" (\(\begin{aligned} \text{User user}\) -> user.id > \(\text{0}\)\)
51
         >>= V.exceptCond "User name must not be empty" (\(User user) -> user.name /= "")
52
   updateUser :: forall eff. PG.Pool -> Handler (postgreSQL :: PG.POSTGRESQL | eff)
   updateUser pool = V.withValidation (Tuple <$> getUserId <*> getUserPatch)
                                       \(Tuple userId (UserPatch userPatch)) ->
56
57
       case unNullOrUndefined userPatch.name of
         Nothing -> respondNoContent 204
59
         Just uName -> V.withValidation (getUserName uName) \userName -> do
60
           savedUser <- liftAff $ PG.withConnection pool \conn -> PG.withTransaction conn
   do
61
             P.findUser conn userId >>= case of
               Nothing -> pure Nothing
63
               Just (User user) -> do
                 let user' = User (user { name = userName })
64
                 P.updateUser conn user'
65
```

```
pure $ Just user'
66
           case savedUser of
67
             Nothing -> respond 404 { error: "User not found with id: " <> show userId }
68
             Just user -> respond 200 (encode user)
69
70
     where
71
       getUserPatch = lift getBody >>= V.except <<< renderForeignErrors</pre>
       getUserName = V.exceptCond "User name must not be empty" (_ /= "")
72
73
74 listUsers :: forall eff. PG.Pool -> Handler (postgreSQL :: PG.POSTGRESQL | eff)
75 listUsers pool = liftAff (PG.withConnection pool P.listUsers) >>= encode >>> respond 200
76
77 getUserId :: forall eff. V. Validation eff Int
   getUserId = lift (getRouteParam "id")
     >>= V.exceptMaybe "User ID is required"
79
     >>= fromString >>> V.exceptMaybe "User ID must be an integer"
     >>= V.exceptCond "User ID must be positive" (_ > 0)
81
82
83 renderForeignErrors :: forall a. Either (NonEmptyList ForeignError) a -> Either String a
   renderForeignErrors = lmap (toList >>> map renderForeignError >>> intercalate ", ")
85
86 respond :: forall eff a. Int -> a -> Handler eff
87 respond status body = do
     setStatus status
89
     sendJson body
90
91 respondNoContent :: forall eff. Int -> Handler eff
92 respondNoContent status = do
     setStatus status
93
94
     end
```

The code is much cleaner now. Let's try out a few test cases:

```
1 $ http POST http://localhost:4000/v1/users id:=3 name=roger
2 HTTP/1.1 201 Created
3 Connection: keep-alive
4 Content-Length: 0
5 Date: Sat, 30 Sep 2017 12:13:37 GMT
6 X-Powered-By: Express
1  $ http POST http://localhost:4000/v1/users id:=3
2 HTTP/1.1 422 Unprocessable Entity
3 Connection: keep-alive
4 Content-Length: 102
5 Content-Type: application/json; charset=utf-8
6 Date: Sat, 30 Sep 2017 12:13:50 GMT
7 ETag: W/"66-/c4cfoquQZGwtDBUzHjJydJAHJ0"
8 X-Powered-By: Express
9
10 {
11
       "error": "Error at array index 0: (ErrorAtProperty \"name\" (TypeMismatch \"String\"
```

```
\"Undefined\"))"
12 }
1 $ http POST http://localhost:4000/v1/users id:=3 name=""
2 HTTP/1.1 422 Unprocessable Entity
3 Connection: keep-alive
4 Content-Length: 39
5 Content-Type: application/json; charset=utf-8
6 Date: Sat, 30 Sep 2017 12:14:02 GMT
7 ETag: W/"27-JQsh12xu/rEFdWy8REF4NMtBUB4"
8 X-Powered-By: Express
9
10 {
11
      "error": "User name must not be empty"
12 }
1 $ http POST http://localhost:4000/v1/users id:=0 name=roger
2 HTTP/1.1 422 Unprocessable Entity
3 Connection: keep-alive
4 Content-Length: 36
5 Content-Type: application/json; charset=utf-8
6 Date: Sat, 30 Sep 2017 12:14:14 GMT
7 ETag: W/"24-Pvt1L4eGilBmVtaOGHlSReJ413E"
8 X-Powered-By: Express
9
10 {
11
      "error": "User ID must be positive"
12 }
1  $ http GET http://localhost:4000/v1/user/3
2 HTTP/1.1 200 OK
3 Connection: keep-alive
4 Content-Length: 23
5 Content-Type: application/json; charset=utf-8
6 Date: Sat, 30 Sep 2017 12:14:28 GMT
7 ETag: W/"17-1scpiB1FT9DBu9s4I1gNWSjH2go"
8 X-Powered-By: Express
10 {
      "id": 3,
12
       "name": "roger"
13 }
1 $ http GET http://localhost:4000/v1/user/asdf
2 HTTP/1.1 422 Unprocessable Entity
3 Connection: keep-alive
4 Content-Length: 38
5 Content-Type: application/json; charset=utf-8
6 Date: Sat, 30 Sep 2017 12:14:40 GMT
```

```
ETag: W/"26-//tvORl1gGDUMwgSaqbEpJhuadI"
  X-Powered-By: Express
10 {
11
       "error": "User ID must be an integer"
  $ http GET http://localhost:4000/v1/user/-1
2 HTTP/1.1 422 Unprocessable Entity
3 Connection: keep-alive
4 Content-Length: 36
5 Content-Type: application/json; charset=utf-8
6 Date: Sat, 30 Sep 2017 12:14:45 GMT
7 ETag: W/"24-Pvt1L4eGilBmVtaOGHlSReJ413E"
 X-Powered-By: Express
10 {
       "error": "User ID must be positive"
12 }
```

It works as expected.

Configuration

Right now our application configuration resides in the main function:

```
main = runServer port databaseConfig
2
     where
       port = 4000
3
       databaseConfig = { user: "abhinav"
4
                         , password: ""
5
                         , host: "localhost"
7
                         , port: 5432
                         , database: "simple_service"
8
                         , max: 10
                           idleTimeoutMillis: 1000
10
```

We are going to extract it out of the code and read it from the environment variables using the purescript-config package. First, we install the required packages using bower.

```
1 $ bower install --save purescript-node-process purescript-config
```

Now, we write the following code in the src/SimpleService/Config.purs file:

```
1 module SimpleService.Config where
2
3 import Data.Config
4 import Prelude
```

```
5
   import Control.Monad.Eff (Eff)
   import Data.Config.Node (fromEnv)
 import Data.Either (Either)
   import Data.Set (Set)
10 import Database.PostgreSQL as PG
   import Node.Process (PROCESS)
12
13 type ServerConfig =
14
     { port
                      :: Int
     , databaseConfig :: PG.PoolConfiguration
15
16
17
18 databaseConfig :: Config {name :: String} PG.PoolConfiguration
   databaseConfig =
19
     { user: _, password: _, host: _, port: _, database: _, max: _, idleTimeoutMillis: _ }
20
     <$> string {name: "user"}
21
     <*> string {name: "password"}
22
     <*> string {name: "host"}
23
     <*> int {name: "port"}
24
     <*> string {name: "database"}
25
     <*> int {name: "pool_size"}
26
     <*> int
               {name: "idle_conn_timeout_millis"}
27
28
29 portConfig :: Config {name :: String} Int
   portConfig = int {name: "port"}
31
32 serverConfig :: Config {name :: String} ServerConfig
33 serverConfig =
     { port: _, databaseConfig: _}
34
     <$> portConfig
35
     <*> prefix {name: "db"} databaseConfig
36
38 readServerConfig :: forall eff.
                       Eff (process :: PROCESS | eff) (Either (Set String) ServerConfig)
39
40 readServerConfig = fromEnv "SS" serverConfig
```

We use the applicative DSL provided in Data.Config module to build a description of our configuration. This description contains the keys and types of the configuration, for consumption by various interpreters. Then we use the fromEnv interpreter to read the config from the environment variables derived from the name fields in the records in the description in the readServerConfig function. We also write a bash script to set those environment variables in the development environment in the setenv.sh file:

```
1 export SS_PORT=4000
2 export SS_DB_USER="abhinav"
3 export SS_DB_PASSWORD=""
4 export SS_DB_HOST="localhost"
5 export SS_DB_PORT=5432
6 export SS_DB_DATABASE="simple_service"
7 export SS_DB_POOL_SIZE=10
```

```
8 export SS_DB_IDLE_CONN_TIMEOUT_MILLIS=1000
```

Now we rewrite our src/Main.purs file to use the readServerConfig function:

```
module Main where
  import Prelude
3
5 import Control.Monad.Eff (Eff)
6 import Control.Monad.Eff.Console (CONSOLE, log)
7 import Data.Either (Either(..))
8 import Data.Set (toUnfoldable)
9 import Data.String (joinWith)
10 import Database.PostgreSQL as PG
11 import Node.Express.Types (EXPRESS)
12 import Node.Process (PROCESS)
13 import Node.Process as Process
14 import SimpleService.Config (readServerConfig)
15 import SimpleService.Server (runServer)
   main :: forall eff. Eff ( console :: CONSOLE
17
                           , express :: EXPRESS
18
19
                            , postgreSQL :: PG.POSTGRESQL
                            , process :: PROCESS
20
                           | eff ) Unit
21
22 main = readServerConfig >>= case _ of
     Left missingKeys -> do
       log $ "Unable to start. Missing Env keys: " <> joinWith ", " (toUnfoldable
   missingKeys)
25
       Process.exit 1
     Right { port, databaseConfig } -> runServer port databaseConfig
26
```

If readServerConfig fails, we print the missing keys to the console and exit the process. Else we run the server with the read config.

To test this, we stop the server we ran in the beginning, source the config, and run it again:

```
1  $ pulp --watch run
2  * Building project in /Users/abhinav/ps-simple-rest-service
3  * Build successful.
4  Server listening on :4000
5  ^C
6  $ source setenv.sh
7  $ pulp --watch run
8  * Building project in /Users/abhinav/ps-simple-rest-service
9  * Build successful.
10  Server listening on :4000
```

It works! We test the failure case by opening another terminal which does not have the environment variables set:

Up next, we add logging to our application.

Logging

For logging, we use the purescript-logging package. We write a logger which logs to stdout; in the src/SimpleService/Logger.purs file:

```
module SimpleService.Logger
    ( debug
2
3
     , info
     , warn
5
     , error
     ) where
7
8
 import Prelude
10 import Control.Logger as L
import Control.Monad.Eff.Class (class MonadEff, liftEff)
12 import Control.Monad.Eff.Console as C
13 import Control.Monad.Eff.Now (NOW, now)
14 import Data.DateTime.Instant (toDateTime)
15 import Data.Either (fromRight)
16 import Data.Formatter.DateTime (Formatter, format, parseFormatString)
17 import Data.Generic.Rep (class Generic)
18 import Data.Generic.Rep.Show (genericShow)
19 import Data.String (toUpper)
20 import Partial.Unsafe (unsafePartial)
22 data Level = Debug | Info | Warn | Error
23
24 derive instance eqLevel :: Eq Level
25 derive instance ordLevel :: Ord Level
26 derive instance genericLevel :: Generic Level _
27
28 instance showLevel :: Show Level where
29
     show = toUpper <<< genericShow</pre>
30
31 type Entry =
   { level :: Level
32
     , message :: String
33
34
     }
35
```

```
36 dtFormatter :: Formatter
   dtFormatter = unsafePartial $ fromRight $ parseFormatString "YYYY-MM-DD HH:mm:ss.SSS"
39 logger :: forall m e. (
40
             MonadEff (console :: C.CONSOLE, now :: NOW | e) m) => L.Logger m Entry
41 logger = L.Logger $ \{ level, message } -> liftEff do
     time <- toDateTime <$> now
42
     C.log $ "[" <> format dtFormatter time <> "] " <> show level <> " " <> message
43
44
45 log :: forall m e.
46
           MonadEff (console :: C.CONSOLE , now :: NOW | e) m
        => Entry -> m Unit
47
   log entry@{level} = L.log (L.cfilter (\e -> e.level == level) logger) entry
49
   debug :: forall m e.
51
            MonadEff (console :: C.CONSOLE , now :: NOW | e) m => String -> m Unit
   debug message = log { level: Debug, message }
54 info :: forall m e.
           MonadEff (console :: C.CONSOLE , now :: NOW | e) m => String -> m Unit
56 info message = log { level: Info, message }
57
58 warn :: forall m e.
           MonadEff (console :: C.CONSOLE , now :: NOW | e) m => String -> m Unit
59
60 warn message = log { level: Warn, message }
61
62 error :: forall m e.
            MonadEff (console :: C.CONSOLE , now :: NOW | e) m => String -> m Unit
64 error message = log { level: Error, message }
```

entry type with the log level and the message. Then we write the logger which will print the log entry to stdout along with the current time as a well formatted string. We define convenience functions for each log level.

Before we proceed, let's install the required dependencies.

```
1 $ bower install --save purescript-logging purescript-now purescript-formatters
```

Now we add a request logger middleware to our server in the src/SimpleService/Server.purs file:

```
1  -- previous code
2  import Control.Monad.Eff.Console (CONSOLE)
3  import Control.Monad.Eff.Now (NOW)
4  import Data.Maybe (maybe)
5  import Data.String (toUpper)
6  import Node.Express.App (App, all, delete, get, http, listenHttp, post, use, useExternal, useOnError)
7  import Node.Express.Handler (Handler, next)
8  import Node.Express.Request (getMethod, getPath)
```

```
import SimpleService.Logger as Log
10 -- previous code
12 requestLogger :: forall eff. Handler (console :: CONSOLE, now :: NOW | eff)
13 requestLogger = do
     method <- getMethod
14
          <- getPath
     path
15
     Log.debug $ "HTTP: " <> maybe "" id ((toUpper <<< show) <$> method) <> " " <> path
17
     next
18
19 app :: forall eff.
          PG.Pool
20
       -> App (postgreSQL :: PG.POSTGRESQL, console :: CONSOLE, now :: NOW | eff)
22 app pool = do
     useExternal jsonBodyParser
23
24
     use requestLogger
25
     -- previous code
```

We also convert all our previous logging statements which used Console.log to use SimpleService.Logger and add logs in our handlers. We can see logging in effect by restarting the server and hitting it:

```
1 $ pulp --watch run
2 * Building project in /Users/abhinav/ps-simple-rest-service
3 * Build successful.
4 [2017-09-30 16:02:41.634] INFO Server listening on :4000
5 [2017-09-30 16:02:43.494] DEBUG HTTP: PATCH /v1/user/3
6 [2017-09-30 16:02:43.517] DEBUG Updated user: 3
7 [2017-09-30 16:03:46.615] DEBUG HTTP: DELETE /v1/user/3
8 [2017-09-30 16:03:46.635] DEBUG Deleted user 3
9 [2017-09-30 16:05:03.805] DEBUG HTTP: GET /v1/users
```

Conclusion

In this tutorial we learned how to create a simple JSON REST web service written in PureScript with persistence, validation, configuration and logging. The complete code for this tutorial can be found in github. This post can be discussed on r/purescript.

© 2017, Abhinav Sarkar Generated by Hakyll | Source