

Fast Sudoku Solver in Haskell #1: A Simple Solution

✍ Posted on June 28, 2018 by Abhinav Sarkar

🔖 Tags: haskell, sudoku, programming, puzzle

31 minute read

Sudoku is a number placement puzzle. It consists of a 9x9 grid which is to be filled with digits from 1 to 9 such that each row, each column and each of the nine 3x3 sub-grids contain all of the digits. Some of the cells of the grid come pre-filled and the player has to fill the rest.

Haskell is a purely functional programming language and is a good choice to solve Sudoku given the problem's combinatorial nature. In fact, a lot of people have solved Sudoku in Haskell already and some of those solutions are listed on the Haskell Wiki. The aim of this series of posts is to write a **fast** Sudoku solver in Haskell, so we'll focus on both implementing the solution and making it efficient, step-by-step, starting with a slow but simple solution in this post¹.

This post can be discussed on r/haskell.

Contents

1. Constraint Satisfaction Problem
2. Setting up
3. Pruning the Cells
4. Pruning the Grid
5. Making the Choice
6. Solving the Puzzle
7. Conclusion

Constraint Satisfaction Problem

Solving Sudoku is a constraint satisfaction problem. We are given a partially filled grid and we have to fill the rest of the cells such that each of the following constraints are satisfied:

1. Each of the nine rows must have all of the digits, from 1 to 9.
2. Each of the nine columns must have all of the digits, from 1 to 9.
3. Each of the nine 3x3 sub-grids must have all of the digits, from 1 to 9.

| | | | | | | | | |
|---|---|---|---|---|---|---|---|---|
| . | . | . | . | . | . | . | 1 | . |
| 4 | . | . | . | . | . | . | . | . |
| . | 2 | . | . | . | . | . | . | . |
| . | . | . | . | 5 | . | 4 | . | 7 |
| . | . | 8 | . | . | . | 3 | . | . |
| . | . | 1 | . | 9 | . | . | . | . |
| 3 | . | . | 4 | . | . | 2 | . | . |
| . | 5 | . | 1 | . | . | . | . | . |
| . | . | . | 8 | . | 6 | . | . | . |

A sample puzzle

| | | | | | | | | |
|---|---|---|---|---|---|---|---|---|
| 6 | 9 | 3 | 7 | 8 | 4 | 5 | 1 | 2 |
| 4 | 8 | 7 | 5 | 1 | 2 | 9 | 3 | 6 |
| 1 | 2 | 5 | 9 | 6 | 3 | 8 | 7 | 4 |
| 9 | 3 | 2 | 6 | 5 | 1 | 4 | 8 | 7 |
| 5 | 6 | 8 | 2 | 4 | 7 | 3 | 9 | 1 |
| 7 | 4 | 1 | 3 | 9 | 8 | 6 | 2 | 5 |
| 3 | 1 | 9 | 4 | 7 | 5 | 2 | 6 | 8 |
| 8 | 5 | 6 | 1 | 2 | 9 | 7 | 4 | 3 |
| 2 | 7 | 4 | 8 | 3 | 6 | 1 | 5 | 9 |

and its solution

Each cell in the grid is a member of one row, one column and one sub-grid (called “block” in general). Digits in the pre-filled cells impose constraints on the row, column, and sub-grids they are part of. For example, if a cell contains “1” then no other cell in that cell’s row, column or sub-grid can contain “1”. Given these constraints, we can devise a simple algorithm to solve Sudoku:

1. Each cell contains either a single digit or has a set of possible digits. For example, a grid showing the possibilities of all non-filled cells for the sample puzzle above:

| | | | | | | | | |
|-------------|-------------|-------------|-------------|-------------|-------------|-------------|-------------|-------------|
| [123456789] | [123456789] | [123456789] | [123456789] | [123456789] | [123456789] | [123456789] | 1 | [123456789] |
| 4 | [123456789] | [123456789] | [123456789] | [123456789] | [123456789] | [123456789] | [123456789] | [123456789] |
| [123456789] | 2 | [123456789] | [123456789] | [123456789] | [123456789] | [123456789] | [123456789] | [123456789] |
| [123456789] | [123456789] | [123456789] | [123456789] | 5 | [123456789] | [123456789] | 4 | [123456789] |
| [123456789] | [123456789] | 8 | [123456789] | [123456789] | [123456789] | [123456789] | 3 | [123456789] |
| [123456789] | [123456789] | 1 | [123456789] | 9 | [123456789] | [123456789] | [123456789] | [123456789] |
| 3 | [123456789] | [123456789] | 4 | [123456789] | [123456789] | 2 | [123456789] | [123456789] |
| [123456789] | 5 | [123456789] | 1 | [123456789] | [123456789] | [123456789] | [123456789] | [123456789] |
| [123456789] | [123456789] | [123456789] | 8 | [123456789] | 6 | [123456789] | [123456789] | [123456789] |

2. If a cell contains a digit, remove that digit from the list of the possible digits from all its neighboring cells, where neighboring cells are defined as the other cells in the given cell’s row, column and sub-grid. For example, the grid after removing the fixed value “4” of the row-2-column-1 cell from its neighboring cells:

| | | | | | | | | |
|-------------|-------------|-------------|-------------|-------------|-------------|-------------|-------------|-------------|
| [123 56789] | [123 56789] | [123 56789] | [123456789] | [123456789] | [123456789] | [123456789] | 1 | [123456789] |
| 4 | [123 56789] | [123 56789] | [123 56789] | [123 56789] | [123 56789] | [123 56789] | [123 56789] | [123 56789] |
| [123 56789] | 2 | [123 56789] | [123456789] | [123456789] | [123456789] | [123456789] | [123456789] | [123456789] |
| [123 56789] | [123456789] | [123456789] | [123456789] | 5 | [123456789] | [123456789] | 4 | [123456789] |
| [123 56789] | [123456789] | 8 | [123456789] | [123456789] | [123456789] | [123456789] | 3 | [123456789] |
| [123 56789] | [123456789] | 1 | [123456789] | 9 | [123456789] | [123456789] | [123456789] | [123456789] |
| 3 | [123456789] | [123456789] | 4 | [123456789] | [123456789] | 2 | [123456789] | [123456789] |
| [123 56789] | 5 | [123456789] | 1 | [123456789] | [123456789] | [123456789] | [123456789] | [123456789] |
| [123 56789] | [123456789] | [123456789] | 8 | [123456789] | 6 | [123456789] | [123456789] | [123456789] |

3. Repeat the previous step for all the cells that are have been solved (or “fixed”), either pre-filled or filled in the previous iteration of the solution. For example, the grid after removing all fixed values from all non-fixed cells:

| | | | | | | | |
|-------------|-------------|------------|-------------|-------------|-------------|----------------------|--------------------|
| [56789] | [3 6789] | [3 567 9] | [23 567 9] | [234 678] | [2345 789] | [56789] 1 | [23456 89] |
| 4 | [1 3 6789] | [3 567 9] | [23 567 9] | [123 678] | [123 5 789] | [56789] [23 56789] | [23 56 89] |
| [1 56789] 2 | [3 567 9] | [3 567 9] | [3 567 9] | [1 34 678] | [1 345 789] | [56789] [3456789] | [3456 89] |
| [2 6 9] | [3 6 9] | [23 6 9] | [23 6] | 5 | [123 8] | 4 | [2 6 89] 7 |
| [2 567 9] | [4 67 9] | 8 | [2 67] | [12 4 67] | [12 4 7] | 3 | [2 56 9] |
| [2 567] | [34 67] | 1 | [23 67] | 9 | [234 78] | [56 8] | [2 56 8] |
| 3 | [1 6789] | [67 9] | 4 | 7 | [5 7 9] | 2 | [56789] [1 56 89] |
| [2 6789] 5 | [2 4 67 9] | 1 | [23 7] | [23 7 9] | [23 7 9] | [6789] [34 6789] | [34 6 89] |
| [12 7 9] | [1 4 7 9] | [2 4 7 9] | 8 | [23 7] | 6 | [1 5 7 9] | [345 7 9] |

4. Continue till the grid “settles”, that is, there are no more changes in the possibilities of any cells. For example, the settled grid for the current iteration:

| | | | | | | | |
|-------------|-------------|------------|-------------|-------------|-------------|----------------------|--------------------|
| [56789] | [3 6789] | [3 567 9] | [23 567 9] | [234 6 8] | [2345 789] | [56789] 1 | [23456 89] |
| 4 | [1 3 6789] | [3 567 9] | [23 567 9] | [123 6 8] | [123 5 789] | [56789] [23 56789] | [23 56 89] |
| [1 56789] 2 | [3 567 9] | [3 567 9] | [3 567 9] | [1 34 6 8] | [1 345 789] | [56789] [3456789] | [3456 89] |
| [2 6 9] | [3 6 9] | [23 6 9] | [23 6] | 5 | [123 8] | 4 | [2 6 89] 7 |
| [2 567 9] | [4 67 9] | 8 | [2 67] | [12 4 6] | [12 4 7] | 3 | [2 56 9] |
| [2 567] | [34 67] | 1 | [23 67] | 9 | [234 78] | [56 8] | [2 56 8] |
| 3 | [1 6 89] | [6 9] | 4 | 7 | [5 9] | 2 | [56 89] [1 56 89] |
| [2 6789] 5 | [2 4 67 9] | 1 | [23] | [23 9] | [23 9] | [6789] [34 6789] | [34 6 89] |
| [12 7 9] | [1 4 7 9] | [2 4 7 9] | 8 | [23] | 6 | [1 5 7 9] | [345 7 9] |

5. Once the grid settles, choose one of the non-fixed cells, either randomly or following some strategy. Select one of the digits from all the possibilities of the cell, and fix (assume) the cell to have that digit. Go back to step 1 and repeat.
6. If the elimination of possibilities results in an inconsistency — for example, you end up with a cell with no possibilities — discard that branch of solution, and backtrack to last point where you fixed a cell. Choose a different possibility to fix and repeat.
7. If at any point the grid is completely filled, you’ve found the solution!
8. If you exhaust all branches of the solution then the puzzle is unsolvable. This can happen if it starts with cells wrongly pre-filled.

This algorithm is actually a Depth-First Search on the state space of the grid configurations, which is guaranteed to either find a solution or prove a puzzle to be unsolvable.

Setting up

We start with writing types to represent the cells and the grid.

```
1 data Cell = Fixed Int | Possible [Int] deriving (Show, Eq)
2 type Row  = [Cell]
3 type Grid = [Row]
```

Since a cell is either fixed with a particular digit or has a set of digits as possibilities, so it is natural to represent it as a sum type with `Fixed` and `Possible` constructors. A row is a list of cells and a grid is a list of rows.

We’ll take the input puzzle as a string of 81 characters representing the cells, left-to-right and top-to-bottom. An example is:

```
.....1.4.....2.....5.4.7..8...3...1.9....3..4..2...5.1.....8.6...
```

Here, `.` represents an non-filled cell. Let’s write a function to read this input and parse it to our `Grid` data structure:

```

1 readGrid :: String -> Maybe Grid
2 readGrid s
3   | length s == 81 = traverse (traverse readCell) . Data.List.Split.chunksOf 9 $ s
4   | otherwise      = Nothing
5   where
6     readCell '.' = Just $ Possible [1..9]
7     readCell c
8       | Data.Char.isDigit c && c > '0' = Just . Fixed . Data.Char.digitToInt $ c
9       | otherwise = Nothing

```

readGrid return a Just grid if the input is correct, else it returns a Nothing. It parses a . to a Possible cell with all digits as possibilities, and a digit char to a Fixed cell with that digit. Let's try it out in the *REPL*:

```

1 *Main> Just grid = readGrid ".....1.4.....2.....5.4.7..8...3...1.9....3..4..2...5.1.....8.6..."
2 *Main> mapM_ print grid
3 [Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Possible
4  [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Fixed
5  1,Possible [1,2,3,4,5,6,7,8,9]]
6 [Fixed 4,Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Possible
7  [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Possible
8  [1,2,3,4,5,6,7,8,9]]
9 [Possible [1,2,3,4,5,6,7,8,9],Fixed 2,Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Possible
10 [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Possible
11 [1,2,3,4,5,6,7,8,9]]
12 [Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Possible
13 [1,2,3,4,5,6,7,8,9],Fixed 5,Possible [1,2,3,4,5,6,7,8,9],Fixed 4,Possible [1,2,3,4,5,6,7,8,9],Fixed 7]
14 [Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Fixed 8,Possible [1,2,3,4,5,6,7,8,9],Possible
15 [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Fixed 3,Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9]]
16 [Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Fixed 1,Possible [1,2,3,4,5,6,7,8,9],Fixed 9,Possible
17 [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9]]
18 [Fixed 3,Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Fixed 4,Possible [1,2,3,4,5,6,7,8,9],Possible
19 [1,2,3,4,5,6,7,8,9],Fixed 2,Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9]]
20 [Possible [1,2,3,4,5,6,7,8,9],Fixed 5,Possible [1,2,3,4,5,6,7,8,9],Fixed 1,Possible [1,2,3,4,5,6,7,8,9],Possible
21 [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9]]
22 [Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Fixed 8,Possible
23 [1,2,3,4,5,6,7,8,9],Fixed 6,Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9],Possible [1,2,3,4,5,6,7,8,9]]

```

The output is a bit unreadable but correct. We can write a few functions to clean it up:

```

1  showGrid :: Grid -> String
2  showGrid = unlines . map (unwords . map showCell)
3      where
4          showCell (Fixed x) = show x
5          showCell _ = "."
6
7  showGridWithPossibilities :: Grid -> String
8  showGridWithPossibilities = unlines . map (unwords . map showCell)
9      where
10         showCell (Fixed x)      = show x ++ "          "
11         showCell (Possible xs) =
12             ( ++ "]" )
13             . Data.List.foldl' (\acc x -> acc ++ if x `elem` xs then show x else " ") "["
14             $ [1..9]

```

Back to the *REPL* again:

```

1  *Main> Just grid = readGrid
    ".....1.4.....2.....5.4.7..8...3....1.9....3..4..2...5.1.....8.6..."
2  *Main> putStrLn $ showGrid grid
3  . . . . . 1 .
4  4 . . . . .
5  . 2 . . . . .
6  . . . 5 . 4 . 7
7  . . 8 . . . 3 . .
8  . . 1 . 9 . . . .
9  3 . . 4 . . 2 . .
10 . 5 . 1 . . . . .
11 . . . 8 . 6 . . .

1  *Main> putStrLn $ showGridWithPossibilities grid
2  [123456789] [123456789] [123456789] [123456789] [123456789] [123456789] [123456789] 1 [123456789]
3  4 [123456789] [123456789] [123456789] [123456789] [123456789] [123456789] [123456789] [123456789]
4  [123456789] 2 [123456789] [123456789] [123456789] [123456789] [123456789] [123456789] [123456789]
5  [123456789] [123456789] [123456789] [123456789] 5 [123456789] 4 [123456789] 7
6  [123456789] [123456789] 8 [123456789] [123456789] [123456789] 3 [123456789] [123456789]
7  [123456789] [123456789] 1 [123456789] 9 [123456789] [123456789] [123456789] [123456789]
8  3 [123456789] [123456789] 4 [123456789] [123456789] 2 [123456789] [123456789]
9  [123456789] 5 [123456789] 1 [123456789] [123456789] [123456789] [123456789] [123456789]
10 [123456789] [123456789] [123456789] 8 [123456789] 6 [123456789] [123456789] [123456789]

```

The output is more readable now. We see that, at the start, all the non-filled cells have all digits as possible values. We'll use these functions for debugging as we go forward. We can now start solving the puzzle.

Pruning the Cells

Instead of removing fixed digits of a cell from its neighboring cells, one cell at a time, it is easier to find all the fixed digits in a row of cells and remove all of them from the possibilities of all the non-fixed cells of the

row. Then we can repeat this “pruning” step for all the rows of the grid (and columns and sub-grids too! We’ll see how).

```
1 pruneCells :: [Cell] -> Maybe [Cell]
2 pruneCells cells = traverse pruneCell cells
3   where
4     fixeds = [x | Fixed x <- cells]
5
6     pruneCell (Possible xs) = case xs Data.List.\ fixeds of
7       [] -> Nothing
8       [y] -> Just $ Fixed y
9       ys -> Just $ Possible ys
10    pruneCell x = Just x
```

`pruneCells` prunes a list of cells as described previously. We start with finding the fixed digits in the list of cells. Then we go over each non-fixed cells, removing the fixed digits we found from their possible values. Two special cases arise:

- If pruning results in a cell with no possible digits, it is a sign that this branch of search has no solution and hence, we return a `Nothing` in that case.
- If only one possible digit remains after pruning, then we turn that cell into a fixed cell with that digit.

We use the `traverse` function for pruning the cells so that a `Nothing` resulting from pruning one cell propagates to the entire list.

Let’s take it for a spin in the *REPL*:

```
1 *Main> Just grid = readGrid "6.....1.4.....2.....5.4.7..8...3....1.9....3..4..2...5.1.....8.6..."
2 *Main> putStr $ showGridWithPossibilities $ [head grid] -- first row of the grid
3 6      [123456789] [123456789] [123456789] [123456789] [123456789] [123456789] 1      [123456789]
4 *Main> putStr $ showGridWithPossibilities [fromJust $ pruneCells $ head grid] -- same row after pruning
5 6      [ 2345 789] [ 2345 789] [ 2345 789] [ 2345 789] [ 2345 789] [ 2345 789] 1      [ 2345 789]
```

It works! 6 and 1 are removed from the possibilities of the other cells. Now we are ready for ...

Pruning the Grid

Pruning a grid requires us to prune each row, each column and each sub-grid. Let’s try to solve it in the *REPL* first:

```

1 *Main> Just grid = readGrid "6.....1.4.....2.....5.4.7..8...3....1.9....3..4..2...5.1.....8.6..."
2 *Main> Just grid' = traverse pruneCells grid
3 *Main> putStr $ showGridWithPossibilities grid'
4 6      [ 2345 789] [ 2345 789] [ 2345 789] [ 2345 789] [ 2345 789] [ 2345 789] 1      [ 2345 789]
5 4      [123 56789] [123 56789] [123 56789] [123 56789] [123 56789] [123 56789] [123 56789] [123 56789]
6 [1 3456789] 2      [1 3456789] [1 3456789] [1 3456789] [1 3456789] [1 3456789] [1 3456789] [1 3456789]
7 [123 6 89] [123 6 89] [123 6 89] [123 6 89] 5      [123 6 89] 4      [123 6 89] 7
8 [12 4567 9] [12 4567 9] 8      [12 4567 9] [12 4567 9] [12 4567 9] 3      [12 4567 9] [12 4567 9]
9 [ 2345678 ] [ 2345678 ] 1      [ 2345678 ] 9      [ 2345678 ] [ 2345678 ] [ 2345678 ] [ 2345678 ]
10 3      [1 56789] [1 56789] 4      [1 56789] [1 56789] 2      [1 56789] [1 56789]
11 [ 234 6789] 5      [ 234 6789] 1      [ 234 6789] [ 234 6789] [ 234 6789] [ 234 6789] [ 234 6789]
12 [12345 7 9] [12345 7 9] [12345 7 9] 8      [12345 7 9] 6      [12345 7 9] [12345 7 9] [12345 7 9]

```

By traverse-ing the grid with `pruneCells`, we are able to prune each row, one-by-one. Since pruning a row doesn't affect another row, we don't have to pass the resulting rows between each pruning step. That is to say, `traverse` is enough for us, we don't need `foldl` here.

How do we do the same thing for columns now? Since our representation for the grid is rows-first, we first need to convert it to a columns-first representation. Luckily, that's what `Data.List.transpose` function does:

```

1 *Main> Just grid = readGrid
   "693784512487512936125963874932651487568247391741398625319475268856129743274836159"
2 *Main> putStr $ showGrid grid
3 6 9 3 7 8 4 5 1 2
4 4 8 7 5 1 2 9 3 6
5 1 2 5 9 6 3 8 7 4
6 9 3 2 6 5 1 4 8 7
7 5 6 8 2 4 7 3 9 1
8 7 4 1 3 9 8 6 2 5
9 3 1 9 4 7 5 2 6 8
10 8 5 6 1 2 9 7 4 3
11 2 7 4 8 3 6 1 5 9
12 *Main> putStr $ showGrid $ Data.List.transpose grid
13 6 4 1 9 5 7 3 8 2
14 9 8 2 3 6 4 1 5 7
15 3 7 5 2 8 1 9 6 4
16 7 5 9 6 2 3 4 1 8
17 8 1 6 5 4 9 7 2 3
18 4 2 3 1 7 8 5 9 6
19 5 9 8 4 3 6 2 7 1
20 1 3 7 8 9 2 6 4 5
21 2 6 4 7 1 5 8 3 9

```

Pruning columns is easy now:

```

1 *Main> Just grid = readGrid "6.....1.4.....2.....5.4.7..8...3....1.9....3..4..2...5.1.....8.6..."
2 *Main> Just grid' = fmap Data.List.transpose . traverse pruneCells . Data.List.transpose $ grid
3 *Main> putStr $ showGridWithPossibilities grid'
4 6          [1 34 6789] [ 234567 9] [ 23 567 9] [1234 678 ] [12345 789] [1 56789] 1          [123456 89]
5 4          [1 34 6789] [ 234567 9] [ 23 567 9] [1234 678 ] [12345 789] [1 56789] [ 23456789] [123456 89]
6 [12 5 789] 2          [ 234567 9] [ 23 567 9] [1234 678 ] [12345 789] [1 56789] [ 23456789] [123456 89]
7 [12 5 789] [1 34 6789] [ 234567 9] [ 23 567 9] 5          [12345 789] 4          [ 23456789] 7
8 [12 5 789] [1 34 6789] 8          [ 23 567 9] [1234 678 ] [12345 789] 3          [ 23456789] [123456 89]
9 [12 5 789] [1 34 6789] 1          [ 23 567 9] 9          [12345 789] [1 56789] [ 23456789] [123456 89]
10 3          [1 34 6789] [ 234567 9] 4          [1234 678 ] [12345 789] 2          [ 23456789] [123456 89]
11 [12 5 789] 5          [ 234567 9] 1          [1234 678 ] [12345 789] [1 56789] [ 23456789] [123456 89]
12 [12 5 789] [1 34 6789] [ 234567 9] 8          [1234 678 ] 6          [1 56789] [ 23456789] [123456 89]

```

We transpose the grid to convert the columns into rows, then prune the rows by traverse-ing `pruneCells` over them, and finally, turn the rows back into columns by transpose-ing the grid back again. The last transpose needs to be `fmap`-ped because `traverse pruneCells` returns a `Maybe`.

Pruning sub-grids is a bit trickier. Following the same idea as pruning columns, we need two functions to transform the sub-grids into rows and back. Let's write the first one:

```

1 subGridsToRows :: Grid -> Grid
2 subGridsToRows =
3   concatMap (\rows -> let [r1, r2, r3] = map (Data.List.Split.chunkOf 3) rows
4                       in zipWith3 (\a b c -> a ++ b ++ c) r1 r2 r3)
5   . Data.List.Split.chunkOf 3

```

And try it out:


```

1  *Main> Just grid = readGrid
    "693784512487512936125963874932651487568247391741398625319475268856129743274836159"
2  *Main> putStr $ showGrid grid
3  6 9 3 7 8 4 5 1 2
4  4 8 7 5 1 2 9 3 6
5  1 2 5 9 6 3 8 7 4
6  9 3 2 6 5 1 4 8 7
7  5 6 8 2 4 7 3 9 1
8  7 4 1 3 9 8 6 2 5
9  3 1 9 4 7 5 2 6 8
10 8 5 6 1 2 9 7 4 3
11 2 7 4 8 3 6 1 5 9
12 *Main> putStr $ showGrid $ subGridsToRows grid
13 6 9 3 4 8 7 1 2 5
14 7 8 4 5 1 2 9 6 3
15 5 1 2 9 3 6 8 7 4
16 9 3 2 5 6 8 7 4 1
17 6 5 1 2 4 7 3 9 8
18 4 8 7 3 9 1 6 2 5
19 3 1 9 8 5 6 2 7 4
20 4 7 5 1 2 9 8 3 6
21 2 6 8 7 4 3 1 5 9

```

You can go over the code and the output and make yourself sure that it works. Also, it turns out that we don't need to write the back-transform function. `subGridsToRows` is its own back-transform:

```

1  *Main> putStr $ showGrid grid
2  6 9 3 7 8 4 5 1 2
3  4 8 7 5 1 2 9 3 6
4  1 2 5 9 6 3 8 7 4
5  9 3 2 6 5 1 4 8 7
6  5 6 8 2 4 7 3 9 1
7  7 4 1 3 9 8 6 2 5
8  3 1 9 4 7 5 2 6 8
9  8 5 6 1 2 9 7 4 3
10 2 7 4 8 3 6 1 5 9
11 *Main> putStr $ showGrid $ subGridsToRows $ subGridsToRows $ grid
12 6 9 3 7 8 4 5 1 2
13 4 8 7 5 1 2 9 3 6
14 1 2 5 9 6 3 8 7 4
15 9 3 2 6 5 1 4 8 7
16 5 6 8 2 4 7 3 9 1
17 7 4 1 3 9 8 6 2 5
18 3 1 9 4 7 5 2 6 8
19 8 5 6 1 2 9 7 4 3
20 2 7 4 8 3 6 1 5 9

```

Nice! Now writing the sub-grid pruning function is easy:

```

1 *Main> Just grid = readGrid "6.....1.4.....2.....5.4.7..8...3....1.9....3..4..2...5.1.....8.6..."
2 *Main> Just grid' = fmap subGridsToRows . traverse pruneCells . subGridsToRows $ grid
3 *Main> putStr $ showGridWithPossibilities grid'
4 6          [1 3 5 789] [1 3 5 789] [123456789] [123456789] [123456789] [ 23456789] 1          [ 23456789]
5 4          [1 3 5 789] [1 3 5 789] [123456789] [123456789] [123456789] [ 23456789] [ 23456789] [ 23456789]
6 [1 3 5 789] 2          [1 3 5 789] [123456789] [123456789] [123456789] [ 23456789] [ 23456789] [ 23456789]
7 [ 234567 9] [ 234567 9] [ 234567 9] [1234 678 ] 5          [1234 678 ] 4          [12 56 89] 7
8 [ 234567 9] [ 234567 9] 8          [1234 678 ] [1234 678 ] [1234 678 ] 3          [12 56 89] [12 56 89]
9 [ 234567 9] [ 234567 9] 1          [1234 678 ] 9          [1234 678 ] [12 56 89] [12 56 89] [12 56 89]
10 3          [12 4 6789] [12 4 6789] 4          [ 23 5 7 9] [ 23 5 7 9] 2          [1 3456789] [1 3456789]
11 [12 4 6789] 5          [12 4 6789] 1          [ 23 5 7 9] [ 23 5 7 9] [1 3456789] [1 3456789] [1 3456789]
12 [12 4 6789] [12 4 6789] [12 4 6789] 8          [ 23 5 7 9] 6          [1 3456789] [1 3456789] [1 3456789]

```

It works well. Now we can string together these three steps to prune the entire grid. We also have to make sure that result of pruning each step is fed into the next step so that fixed cells created into one step cause more pruning in further steps. We use monadic bind (`>=>`) for that. Here's the final code:

```

1 pruneGrid' :: Grid -> Maybe Grid
2 pruneGrid' grid =
3   traverse pruneCells grid
4   >=> fmap Data.List.transpose . traverse pruneCells . Data.List.transpose
5   >=> fmap subGridsToRows . traverse pruneCells . subGridsToRows

```

And the test:

```

1 *Main> Just grid = readGrid "6.....1.4.....2.....5.4.7..8...3....1.9....3..4..2...5.1.....8.6..."
2 *Main> Just grid' = pruneGrid' grid
3 *Main> putStr $ showGridWithPossibilities grid'
4 6      [ 3 789] [ 3 5 7 9] [ 23 5 7 9] [ 234 78 ] [ 2345 789] [ 5 789] 1      [ 2345 89]
5 4      [1 3 789] [ 3 5 7 9] [ 23 567 9] [123 678 ] [123 5 789] [ 56789] [ 23 56789] [ 23 56 89]
6 [1 5 789] 2      [ 3 5 7 9] [ 3 567 9] [1 34 678 ] [1 345 789] [ 56789] [ 3456789] [ 3456 89]
7 [ 2      9] [ 3 6 9] [ 23 6 9] [ 23 6 ] 5      [123 8 ] 4      [ 2 6 89] 7
8 [ 2 5 7 9] [ 4 67 9] 8      [ 2 67 ] [12 4 67 ] [12 4 7 ] 3      [ 2 56 9] [12 56 9]
9 [ 2 5 7 ] [ 34 67 ] 1      [ 23 67 ] 9      [ 234 78 ] [ 56 8 ] [ 2 56 8 ] [ 2 56 8 ]
10 3      [1 6789] [ 67 9] 4      7      [ 5 7 9] 2      [ 56789] [1 56 89]
11 [ 2 789] 5      [ 2 4 67 9] 1      [ 23 7 ] [ 23 7 9] [ 6789] [ 34 6789] [ 34 6 89]
12 [12 7 9] [1 4 7 9] [ 2 4 7 9] 8      [ 23 7 ] 6      [1 5 7 9] [ 345 7 9] [1 345 9]
13 *Main> putStr $ showGrid grid
14 6 . . . . . 1 .
15 4 . . . . .
16 . 2 . . . . .
17 . . . . 5 . 4 . 7
18 . . 8 . . . 3 . .
19 . . 1 . 9 . . . .
20 3 . . 4 . . 2 . .
21 . 5 . 1 . . . . .
22 . . . 8 . 6 . . .
23 *Main> putStr $ showGrid grid'
24 6 . . . . . 1 .
25 4 . . . . .
26 . 2 . . . . .
27 . . . . 5 . 4 . 7
28 . . 8 . . . 3 . .
29 . . 1 . 9 . . . .
30 3 . . 4 7 . 2 . .
31 . 5 . 1 . . . . .
32 . . . 8 . 6 . . .

```

We can clearly see the massive pruning of possibilities all around the grid. We also see a “7” pop up in the row-7-column-5 cell. This means that we can prune the grid further, until it settles. If you are familiar with Haskell, you may recognize this as trying to find a fixed point for the `pruneGrid'` function, except in a monadic context. It's simple to implement:

```

1 pruneGrid :: Grid -> Maybe Grid
2 pruneGrid = fixM pruneGrid'
3   where
4     fixM f x = f x >=> \x' -> if x' == x then return x else fixM f x'

```

The crux of this code is the `fixM` function which takes monadic function `f` and an initial value and recursively calls itself till the return value settles. Let's do another round in the *REPL*:

```

1 *Main> Just grid = readGrid "6.....1.4.....2.....5.4.7..8...3....1.9....3..4..2...5.1.....8.6..."
2 *Main> Just grid' = pruneGrid grid
3 *Main> putStr $ showGridWithPossibilities grid'
4 6      [ 3  789] [ 3 5 7 9] [ 23 5 7 9] [ 234  8 ] [ 2345 789] [  5 789] 1      [ 2345  89]
5 4      [1 3  789] [ 3 5 7 9] [ 23 567 9] [123  6 8 ] [123 5 789] [  56789] [ 23 56789] [ 23 56 89]
6 [1  5 789] 2      [ 3 5 7 9] [ 3 567 9] [1 34 6 8 ] [1 345 789] [  56789] [ 3456789] [ 3456 89]
7 [ 2      9] [ 3  6  9] [ 23  6  9] [ 23  6  ] 5      [123  8 ] 4      [ 2  6 89] 7
8 [ 2 5 7 9] [  4 67 9] 8      [ 2  67 ] [12 4 6  ] [12 4 7  ] 3      [ 2 56 9] [12 56 9]
9 [ 2 5 7  ] [ 34 67  ] 1      [ 23  67 ] 9      [ 234 78 ] [  56 8 ] [ 2 56 8 ] [ 2 56 8 ]
10 3      [1  6 89] [  6  9] 4      7      [  5  9] 2      [  56 89] [1  56 89]
11 [ 2  789] 5      [ 2 4 67 9] 1      [ 23  ] [ 23  9] [  6789] [ 34 6789] [ 34 6 89]
12 [12  7 9] [1  4 7 9] [ 2 4 7 9] 8      [ 23  ] 6      [1  5 7 9] [ 345 7 9] [1 345  9]

```

We see that “7” in the row-7-column-5 cell is eliminated from all its neighboring cells. We can’t prune the grid anymore. Now it’s time to make the choice.

Making the Choice

Once the grid is settled, we need to choose a non-fixed cell and make it fixed by assuming one of its possible values. This gives us two grids, next in the state-space of the solution search: one which has this chosen cell fixed to this chosen digit, and the other in which the chosen cell has all the other possibilities except the one we chose to fix. We call this function `nextGrids`:

```

1 nextGrids :: Grid -> (Grid, Grid)
2 nextGrids grid =
3   let (i, first@(Fixed _), rest) =
4     fixCell
5       . Data.List.minimumBy (compare `Data.Function.on` (possibilityCount . snd))
6       . filter (isPossible . snd)
7       . zip [0..]
8       . concat
9       $ grid
10  in (replace2D i first grid, replace2D i rest grid)
11  where
12    isPossible (Possible _) = True
13    isPossible _           = False
14
15    possibilityCount (Possible xs) = length xs
16    possibilityCount (Fixed _)     = 1
17
18    fixCell (i, Possible [x, y]) = (i, Fixed x, Fixed y)
19    fixCell (i, Possible (x:xs)) = (i, Fixed x, Possible xs)
20    fixCell _                    = error "Impossible case"
21
22    replace2D :: Int -> a -> [[a]] -> [[a]]
23    replace2D i v =
24      let (x, y) = (i `quot` 9, i `mod` 9) in replace x (replace y (const v))
25      replace p f xs = [if i == p then f x else x | (x, i) <- zip xs [0..]]

```

We choose the non-fixed cell with least count of possibilities as the pivot. This strategy make sense intuitively as with a cell with fewest possibilities, we have most chance of being right when assuming one. Fixing a non-fixed cell leads to one of the two cases:

- the cell has only two possible values, resulting in two fixed cells, or,
- the cell has more than two possible values, resulting in one fixed and one non-fixed cell.

Then all we are left with is replacing the non-fixed cell with its fixed and fixed/non-fixed choices, which we do with some math and some list traversal. A quick check on the *REPL*:

```
1 *Main> Just grid = readGrid "6.....1.4.....2.....5.4.7..8...3...1.9....3..4..2...5.1.....8.6..."
2 *Main> Just grid' = pruneGrid grid
3 *Main> putStr $ showGridWithPossibilities grid'
4 6          [ 3 789] [ 3 5 7 9] [ 23 5 7 9] [ 234 8 ] [ 2345 789] [ 5 789] 1          [ 2345 89]
5 4          [1 3 789] [ 3 5 7 9] [ 23 567 9] [123 6 8 ] [123 5 789] [ 56789] [ 23 56789] [ 23 56 89]
6 [1 5 789] 2          [ 3 5 7 9] [ 3 567 9] [1 34 6 8 ] [1 345 789] [ 56789] [ 3456789] [ 3456 89]
7 [ 2 9 ] [ 3 6 9] [ 23 6 9] [ 23 6 ] 5          [123 8 ] 4          [ 2 6 89] 7
8 [ 2 5 7 9] [ 4 67 9] 8          [ 2 67 ] [12 4 6 ] [12 4 7 ] 3          [ 2 56 9] [12 56 9]
9 [ 2 5 7 ] [ 34 67 ] 1          [ 23 67 ] 9          [ 234 78 ] [ 56 8 ] [ 2 56 8 ] [ 2 56 8 ]
10 3          [1 6 89] [ 6 9] 4          7          [ 5 9] 2          [ 56 89] [1 56 89]
11 [ 2 789] 5          [ 2 4 67 9] 1          [ 23 ] [ 23 9] [ 6789] [ 34 6789] [ 34 6 89]
12 [12 7 9] [1 4 7 9] [ 2 4 7 9] 8          [ 23 ] 6          [1 5 7 9] [ 345 7 9] [1 345 9]
13 *Main> -- the row-4-column-1 cell is the first cell with only two possibilities, [2, 9].
14 *Main> -- it is chosen as the pivot cell to find the next grids.
15 *Main> (grid1, grid2) = nextGrids grid'
16 *Main> putStr $ showGridWithPossibilities grid1
17 6          [ 3 789] [ 3 5 7 9] [ 23 5 7 9] [ 234 8 ] [ 2345 789] [ 5 789] 1          [ 2345 89]
18 4          [1 3 789] [ 3 5 7 9] [ 23 567 9] [123 6 8 ] [123 5 789] [ 56789] [ 23 56789] [ 23 56 89]
19 [1 5 789] 2          [ 3 5 7 9] [ 3 567 9] [1 34 6 8 ] [1 345 789] [ 56789] [ 3456789] [ 3456 89]
20 2          [ 3 6 9] [ 23 6 9] [ 23 6 ] 5          [123 8 ] 4          [ 2 6 89] 7
21 [ 2 5 7 9] [ 4 67 9] 8          [ 2 67 ] [12 4 6 ] [12 4 7 ] 3          [ 2 56 9] [12 56 9]
22 [ 2 5 7 ] [ 34 67 ] 1          [ 23 67 ] 9          [ 234 78 ] [ 56 8 ] [ 2 56 8 ] [ 2 56 8 ]
23 3          [1 6 89] [ 6 9] 4          7          [ 5 9] 2          [ 56 89] [1 56 89]
24 [ 2 789] 5          [ 2 4 67 9] 1          [ 23 ] [ 23 9] [ 6789] [ 34 6789] [ 34 6 89]
25 [12 7 9] [1 4 7 9] [ 2 4 7 9] 8          [ 23 ] 6          [1 5 7 9] [ 345 7 9] [1 345 9]
26 *Main> putStr $ showGridWithPossibilities grid2
27 6          [ 3 789] [ 3 5 7 9] [ 23 5 7 9] [ 234 8 ] [ 2345 789] [ 5 789] 1          [ 2345 89]
28 4          [1 3 789] [ 3 5 7 9] [ 23 567 9] [123 6 8 ] [123 5 789] [ 56789] [ 23 56789] [ 23 56 89]
29 [1 5 789] 2          [ 3 5 7 9] [ 3 567 9] [1 34 6 8 ] [1 345 789] [ 56789] [ 3456789] [ 3456 89]
30 9          [ 3 6 9] [ 23 6 9] [ 23 6 ] 5          [123 8 ] 4          [ 2 6 89] 7
31 [ 2 5 7 9] [ 4 67 9] 8          [ 2 67 ] [12 4 6 ] [12 4 7 ] 3          [ 2 56 9] [12 56 9]
32 [ 2 5 7 ] [ 34 67 ] 1          [ 23 67 ] 9          [ 234 78 ] [ 56 8 ] [ 2 56 8 ] [ 2 56 8 ]
33 3          [1 6 89] [ 6 9] 4          7          [ 5 9] 2          [ 56 89] [1 56 89]
34 [ 2 789] 5          [ 2 4 67 9] 1          [ 23 ] [ 23 9] [ 6789] [ 34 6789] [ 34 6 89]
35 [12 7 9] [1 4 7 9] [ 2 4 7 9] 8          [ 23 ] 6          [1 5 7 9] [ 345 7 9] [1 345 9]
```

Solving the Puzzle

We have implemented parts of our algorithm till now. Now we'll put everything together to solve the puzzle. First, we need to know if we are done or have messed up:

```

1  isGridFilled :: Grid -> Bool
2  isGridFilled grid = null [ () | Possible _ <- concat grid ]
3
4  isGridInvalid :: Grid -> Bool
5  isGridInvalid grid =
6      any isInvalidRow grid
7      || any isInvalidRow (Data.List.transpose grid)
8      || any isInvalidRow (subGridsToRows grid)
9      where
10         isInvalidRow row =
11             let fixeds          = [x | Fixed x <- row]
12                 emptyPossibles = [x | Possible x <- row, null x]
13             in hasDups fixeds || not (null emptyPossibles)
14
15         hasDups l = hasDups' l []
16
17         hasDups' [] _ = False
18         hasDups' (y:ys) xs
19             | y `elem` xs = True
20             | otherwise   = hasDups' ys (y:xs)

```

isGridFilled returns whether a grid is filled completely by checking it for any Possible cells. isGridInvalid checks if a grid is invalid because it either has duplicate fixed cells in any block or has any non-fixed cell with no possibilities.

Writing the solve function is now almost trivial:

```

1  solve :: Grid -> Maybe Grid
2  solve grid = pruneGrid grid >>= solve'
3      where
4          solve' g
5              | isGridInvalid g = Nothing
6              | isGridFilled g  = Just g
7              | otherwise       =
8                  let (grid1, grid2) = nextGrids g
9                  in solve grid1 <|> solve grid2

```

We prune the grid as before and pipe it to the helper function solve'. solve' bails with a Nothing if the grid is invalid, or returns the solved grid if it is filled completely. Otherwise, it finds the next two grids in the search tree and solves them recursively with backtracking by calling the solve function. Backtracking here is implemented by using the Alternative (<|>) implementation of the Maybe type (source), which takes the next branch in the computation if the first branch returns a Nothing.

Whew! That took us long. Let's put it to the final test now:

```

1  *Main> Just grid = readGrid
    "6.....1.4.....2.....5.4.7..8...3....1.9....3..4..2...5.1.....8.6..."
2  *Main> putStr $ showGrid grid
3  6 . . . . . 1 .
4  4 . . . . .
5  . 2 . . . . .
6  . . . . 5 . 4 . 7
7  . . 8 . . . 3 . .
8  . . 1 . 9 . . . .
9  3 . . 4 . . 2 . .
10 . 5 . 1 . . . . .
11 . . . 8 . 6 . . .
12 *Main> Just grid' = solve grid
13 *Main> putStr $ showGrid grid'
14 6 9 3 7 8 4 5 1 2
15 4 8 7 5 1 2 9 3 6
16 1 2 5 9 6 3 8 7 4
17 9 3 2 6 5 1 4 8 7
18 5 6 8 2 4 7 3 9 1
19 7 4 1 3 9 8 6 2 5
20 3 1 9 4 7 5 2 6 8
21 8 5 6 1 2 9 7 4 3
22 2 7 4 8 3 6 1 5 9

```

It works! Let's put a quick main wrapper around solve to call it from the command line:

```

1  main :: IO ()
2  main = do
3    inputs <- lines <$> getContents
4    Control.Monad.forM_ inputs $ \input ->
5      case readGrid input of
6        Nothing   -> putStrLn "Invalid input"
7        Just grid -> case solve grid of
8          Nothing   -> putStrLn "No solution found"
9          Just grid' -> putStrLn $ showGrid grid'

```

And now, we can invoke it from the command line:

```

$ echo ".....12.5.4.....3.7..6..4....1.....8....92....8.....51.7.....3..."
| stack exec sudoku
3 6 4 9 7 8 5 1 2
1 5 2 4 3 6 9 7 8
8 7 9 1 2 5 6 3 4
7 3 8 6 5 1 4 2 9
6 9 1 2 4 7 3 8 5
2 4 5 3 8 9 1 6 7
9 2 3 7 6 4 8 5 1
4 8 6 5 1 2 7 9 3
5 1 7 8 9 3 2 4 6

```

And, we are done.

If you want to play with different puzzles, the file here lists some of the toughest ones. Let's run some of them through our program to see how fast it is:

```
$ head -n100 sudoku17.txt | time stack exec sudoku
... output omitted ...
      116.70 real          198.09 user          94.46 sys
```

It took about 292 seconds to solve a hundred puzzles, so, about 3 seconds per puzzle. This is pretty slow but we'll get around to making it faster in the subsequent posts.

Conclusion

In this rather verbose article, we learned how to write a simple Sudoku solver in Haskell step-by-step. In the later parts of this series, we'll delve into profiling the solution and figuring out better algorithms and data structures to solve Sudoku more efficiently. The code till now is available here. This post can be discussed on [r/haskell](#).

1. This exercise was originally done as a part of the Haskell classes I taught at [nilenso](#).¹