# Writing a Simple REST Web Service in PureScript - Part 2

• October 1, 2017
• A twenty minute read
• Tags: purescript, REST, programming, nilenso

o recap, in the first part of this two-part tutorial, we built a simple JSON REST web service in PureScript to create, update, get, list and delete users, backed by a Postgres database. In this part we'll work on the rest of the features. The requirements are:

- 1. validation of API requests.
- 2. reading the server and database configs from environment variables.
- 3. logging HTTP requests and debugging info.

#### **Contents**

- 1. Bugs!
- 2. Validation
- 3. Configuration
- 4. Logging
- 5. Conclusion

But first,

## **Bugs!**

What happens if we hit a URL on our server which does not exist? Let's fire up the server and test it:

1 \$ pulp --watch run

```
$ http GET http://localhost:4000/v1/random
  HTTP/1.1 404 Not Found
3 Connection: keep-alive
4 Content-Length: 148
5 Content-Security-Policy: default-src 'self'
6 Content-Type: text/html; charset=utf-8
7 Date: Sat, 30 Sep 2017 08:23:20 GMT
8 X-Content-Type-Options: nosniff
  X-Powered-By: Express
10
11 <!DOCTYPE html>
12 <html lang="en">
13 <head>
14 <meta charset="utf-8">
15 <title>Error</title>
16 </head>
17 <body>
18 Cannot GET /v1/random
19 </body>
20 </html>
```

We get back a default HTML response with a 404 status from Express. Since we are writing a JSON API, we should return a JSON response in this case too. We add the following code in the src/SimpleService/Server.purs file to add a catch-all route and send a 404 status with a JSON error message:

```
-- previous code
  import Data.Either (fromRight)
  import Data.String.Regex (Regex, regex) as Re
  import Data.String.Regex.Flags (noFlags) as Re
  import Node.Express.App (App, all, delete, get, http, listenHttp, post, useExternal)
  import Node.Express.Response (sendJson, setStatus)
  import Partial.Unsafe (unsafePartial)
  -- previous code
8
10 allRoutePattern :: Re.Regex
11 allRoutePattern = unsafePartial $ fromRight $ Re.regex "/.*" Re.noFlags
13 app :: forall eff. PG.Pool -> App (postgreSQL :: PG.POSTGRESQL | eff)
14 app pool = do
    useExternal jsonBodyParser
16
    17
    delete "/v1/user/:id" $ deleteUser pool
18
    19
    patch "/v1/user/:id" $ updateUser pool
    21
22
    all allRoutePattern do
      setStatus 404
24
      sendJson {error: "Route not found"}
25
    where
26
      patch = http (CustomMethod "patch")
27
```

allRoutePattern matches all routes because it uses a "/.\*" regular expression. We place it as the last route to match all the otherwise unrouted requests. Let's see what is the result:

```
1  $ http GET http://localhost:4000/v1/random
2  HTTP/1.1  404 Not Found
3  Connection: keep-alive
4  Content-Length: 27
5  Content-Type: application/json; charset=utf-8
6  Date: Sat, 30 Sep 2017 08:46:46 GMT
7  ETag: W/"1b-772e0u4nrE48ogbR0KmKfSvrHUE"
8  X-Powered-By: Express
9
10 {
11  "error": "Route not found"
12 }
```

Now we get a nicely formatted JSON response.

Another scenario is when our application throws some uncaught error. To simulate this, we shut down our postgres database and hit the server for listing users:

```
1 $ http GET http://localhost:4000/v1/users
2 HTTP/1.1 500 Internal Server Error
3 Connection: keep-alive
4 Content-Length: 372
5 Content-Security-Policy: default-src 'self'
6 Content-Type: text/html; charset=utf-8
7 Date: Sat, 30 Sep 2017 08:53:40 GMT
8 X-Content-Type-Options: nosniff
9 X-Powered-By: Express
10
11 <!DOCTYPE html>
12 <html lang="en">
13 <head>
14 <meta charset="utf-8">
15 <title>Error</title>
16 </head>
17 <body>
18 Frror: connect ECONNREFUSED 127.0.0.1:5432<br > &nbsp; &nbsp; at
   Object._errnoException (util.js:1026:11)    at _exceptionWithHostPort
   (util.js:1049:20) < br > &nbsp; &nbsp; at TCPConnectWrap.afterConnect [as oncomplete]
   (net.js:1174:14)
19 </body>
20 </html>
```

We get another default HTML response from Express with a 500 status. Again, in this case we'd like to return a JSON response. We add the following code to the src/SimpleService/Server.purs file:

```
1 -- previous code
2 import Control.Monad.Eff.Exception (message)
3 import Node.Express.App (App, all, delete, get, http, listenHttp, post, useExternal,
   useOnError)
4 -- previous code
5
  app :: forall eff. PG.Pool -> App (postgreSQL :: PG.POSTGRESQL | eff)
   app pool = do
    -- previous code
    useOnError \err -> do
9
      setStatus 500
10
       sendJson {error: message err}
12
       patch = http (CustomMethod "patch")
13
```

We add the useOnError handler which comes with purescript-express to return the error message as a JSON response. Back on the command-line:

```
1  $ http GET http://localhost:4000/v1/users
2  HTTP/1.1 500 Internal Server Error
3  Connection: keep-alive
4  Content-Length: 47
5  Content-Type: application/json; charset=utf-8
6  Date: Sat, 30 Sep 2017 09:01:37 GMT
7  ETag: W/"2f-cJuIW6961YCpo9TWDSZ9VWHLGHE"
8  X-Powered-By: Express
9
10 {
11  "error": "connect ECONNREFUSED 127.0.0.1:5432"
12 }
```

It works! Bugs are fixed now. We proceed to add next features.

### **Validation**

Let's recall the code to update a user from the src/SimpleService/Handler.purs file:

```
updateUser :: forall eff. PG.Pool -> Handler (postgreSQL :: PG.POSTGRESQL | eff)
   updateUser pool = getRouteParam "id" >>= case _ of
     Nothing -> respond 422 { error: "User ID is required" }
     Just sUserId -> case fromString sUserId of
       Nothing -> respond 422 { error: "User ID must be positive: " <> sUserId }
       Just userId -> getBody >>= case _ of
         Left errs -> respond 422 { error: intercalate ", " $ map renderForeignError errs}
         Right (UserPatch userPatch) -> case unNullOrUndefined userPatch.name of
           Nothing -> respondNoContent 204
9
           Just userName -> if userName == ""
             then respond 422 { error: "User name must not be empty" }
11
             else do
12
               savedUser <- liftAff $ PG.withConnection pool \conn -> PG.withTransaction
   conn do
                 P.findUser conn userId >>= case _ of
14
                   Nothing -> pure Nothing
15
                   Just (User user) -> do
16
                     let user' = User (user { name = userName })
17
                     P.updateUser conn user'
18
                     pure $ Just user'
19
               case savedUser of
20
                 Nothing -> respond 404 { error: "User not found with id: " <> sUserId }
21
                 Just user -> respond 200 (encode user)
22
```

As we can see, the actual request handling logic is obfuscated by the request validation logic for the user id and the user name patch parameters. We also notice that we are using three constructs for validation here: Maybe, Either and if-then-else. However, we can use just Either to subsume all these cases as it can "carry" a failure as well as a success case. Either also comes with a nice monad transformer ExceptT which provides the do syntax for failure propagation. So we choose ExceptT as the base construct for our validation framework and write functions to upgrade Maybe and if-then-else to it. We add the following code to the src/SimpleService/Validation.purs file:

```
module SimpleService.Validation
1
     (module MoreExports, module SimpleService.Validation) where
2
3
   import Prelude
4
5
   import Control.Monad.Except (ExceptT, except, runExceptT)
6
   import Data.Either (Either(..))
   import Data.Maybe (Maybe(..))
   import Node.Express.Handler (HandlerM, Handler)
import Node.Express.Response (sendJson, setStatus)
  import Node.Express.Types (EXPRESS)
11
import Control.Monad.Except (except) as MoreExports
13
14 type Validation eff a = ExceptT String (HandlerM (express :: EXPRESS | eff)) a
15
16 exceptMaybe :: forall e m a. Applicative m => e -> Maybe a -> ExceptT e m a
  exceptMaybe e a = except $ case a of
17
     Just x -> Right x
18
     Nothing -> Left e
19
20
   exceptCond :: forall e m a. Applicative m => e -> (a -> Boolean) -> a -> ExceptT e m a
   exceptCond e cond a = except $ if cond a then Right a else Left e
22
23
24 withValidation :: forall eff a. Validation eff a -> (a -> Handler eff) -> Handler eff
25 withValidation action handler = runExceptT action >>= case _ of
    Left err -> do
26
       setStatus 422
27
      sendJson {error: err}
28
     Right x \rightarrow \text{handler } x
```

We re-export except from the Control.Monad.Except module. We also add a withValidation function which runs an ExceptT based validation and either returns an error response with a 422 status in case of a failed validation or runs the given action with the valid value in case of a successful validation.

Using these functions, we now write updateUser in the src/SimpleService/Handler.purs file as:

```
-- previous code
   import Control.Monad.Trans.Class (lift)
  import Data.Bifunctor (lmap)
  import Data.Foreign (ForeignError, renderForeignError)
   import Data.List.NonEmpty (toList)
   import Data.List.Types (NonEmptyList)
   import Data.Tuple (Tuple(..))
   import SimpleService. Validation as V
   -- previous code
10
11 renderForeignErrors :: forall a. Either (NonEmptyList ForeignError) a -> Either String a
   renderForeignErrors = lmap (toList >>> map renderForeignError >>> intercalate ", ")
13
   updateUser :: forall eff. PG.Pool -> Handler (postgreSQL :: PG.POSTGRESQL | eff)
   updateUser pool = V.withValidation (Tuple <$> getUserId <*> getUserPatch)
                                       \(Tuple userId (UserPatch userPatch)) ->
16
17
       case unNullOrUndefined userPatch.name of
         Nothing -> respondNoContent 204
18
         Just uName -> V.withValidation (getUserName uName) \userName -> do
19
           savedUser <- liftAff $ PG.withConnection pool \conn -> PG.withTransaction conn
20
   do
             P.findUser conn userId >>= case _ of
21
               Nothing -> pure Nothing
               Just (User user) -> do
23
                 let user' = User (user { name = userName })
24
                 P.updateUser conn user'
25
                 pure $ Just user'
26
           case savedUser of
27
             Nothing -> respond 404 { error: "User not found with id: " <> show userId }
28
             Just user -> respond 200 (encode user)
29
     where
30
       getUserId = lift (getRouteParam "id")
31
         >>= V.exceptMaybe "User ID is required"
32
         >>= fromString >>> V.exceptMaybe "User ID must be positive"
33
34
       getUserPatch = lift getBody >>= V.except <<< renderForeignErrors</pre>
35
36
       getUserName = V.exceptCond "User name must not be empty" (_ == "")
37
```

The validation logic has been extracted out in separate functions now which are composed using Applicative. The validation steps are composed using the ExceptT monad. We are now free to express the core logic of the function clearly. We rewrite the src/SimpleService/Handler.purs file using the validations:

```
module SimpleService. Handler where
  import Prelude
3
4
   import Control.Monad.Aff.Class (liftAff)
5
  import Control.Monad.Trans.Class (lift)
   import Data.Bifunctor (lmap)
   import Data.Either (Either)
   import Data.Foldable (intercalate)
import Data.Foreign (ForeignError, renderForeignError)
import Data.Foreign.Class (encode)
import Data.Foreign.NullOrUndefined (unNullOrUndefined)
import Data.Int (fromString)
14 import Data.List.NonEmpty (toList)
import Data.List.Types (NonEmptyList)
import Data.Maybe (Maybe(..))
import Data.Tuple (Tuple(..))
18 import Database.PostgreSQL as PG
import Node.Express.Handler (Handler)
20 import Node.Express.Request (getBody, getRouteParam)
21 import Node.Express.Response (end, sendJson, setStatus)
22 import SimpleService.Persistence as P
23 import SimpleService. Validation as V
24 import SimpleService.Types
25
26 getUser :: forall eff. PG.Pool -> Handler (postgreSOL :: PG.POSTGRESOL | eff)
  getUser pool = V.withValidation getUserId \userId ->
     liftAff (PG.withConnection pool $ flip P.findUser userId) >>= case _ of
28
       Nothing -> respond 404 { error: "User not found with id: " <> show userId }
29
       Just user -> respond 200 (encode user)
30
32 deleteUser :: forall eff. PG.Pool -> Handler (postgreSQL :: PG.POSTGRESQL | eff)
33 deleteUser pool = V.withValidation getUserId \userId -> do
     found <- liftAff $ PG.withConnection pool \conn -> PG.withTransaction conn do
34
       P.findUser conn userId >>= case _ of
35
         Nothing -> pure false
36
         Just -> do
37
           P.deleteUser conn userId
38
           pure true
    if found
40
       then respondNoContent 204
41
       else respond 404 { error: "User not found with id: " <> show userId }
42
43
44 createUser :: forall eff. PG.Pool -> Handler (postgreSQL :: PG.POSTGRESQL | eff)
45 createUser pool = V.withValidation getUser \user@(User ) -> do
     liftAff (PG.withConnection pool $ flip P.insertUser user)
46
     respondNoContent 201
47
    where
48
     getUser = lift getBody
49
        >>= V.except <<< renderForeignErrors
50
```

```
>>= V.exceptCond "User ID must be positive" (\(User user) -> user.id > 0)
51
         >>= V.exceptCond "User name must not be empty" (\(\begin{align*} \text{User user} \) -> user.name /= "")
53
   updateUser :: forall eff. PG.Pool -> Handler (postgreSQL :: PG.POSTGRESQL | eff)
   updateUser pool = V.withValidation (Tuple <$> getUserId <*> getUserPatch)
                                        \(Tuple userId (UserPatch userPatch)) ->
56
       case unNullOrUndefined userPatch.name of
57
         Nothing -> respondNoContent 204
58
         Just uName -> V.withValidation (getUserName uName) \userName -> do
59
            savedUser <- liftAff $ PG.withConnection pool \conn -> PG.withTransaction conn
   do
             P.findUser conn userId >>= case _ of
61
               Nothing -> pure Nothing
62
               Just (User user) -> do
63
                  let user' = User (user { name = userName })
64
                  P.updateUser conn user'
65
                  pure $ Just user'
66
           case savedUser of
             Nothing -> respond 404 { error: "User not found with id: " <> show userId }
68
              Just user -> respond 200 (encode user)
69
70
     where
       getUserPatch = lift getBody >>= V.except <<< renderForeignErrors</pre>
71
       getUserName = V.exceptCond "User name must not be empty" (_ /= "")
72
73
   listUsers :: forall eff. PG.Pool -> Handler (postgreSQL :: PG.POSTGRESQL | eff)
   listUsers pool = liftAff (PG.withConnection pool P.listUsers) >>= encode >>> respond 200
76
   getUserId :: forall eff. V. Validation eff Int
   getUserId = lift (getRouteParam "id")
     >>= V.exceptMaybe "User ID is required"
79
     >>= fromString >>> V.exceptMaybe "User ID must be an integer"
     >>= V.exceptCond "User ID must be positive" (_ > 0)
81
82
   renderForeignErrors :: forall a. Either (NonEmptyList ForeignError) a -> Either String a
   renderForeignErrors = lmap (toList >>> map renderForeignError >>> intercalate ", ")
84
85
   respond :: forall eff a. Int -> a -> Handler eff
   respond status body = do
87
     setStatus status
88
89
     sendJson body
90
  respondNoContent :: forall eff. Int -> Handler eff
   respondNoContent status = do
     setStatus status
93
     end
```

The code is much cleaner now. Let's try out a few test cases:

```
1 $ http POST http://localhost:4000/v1/users id:=3 name=roger
2 HTTP/1.1 201 Created
3 Connection: keep-alive
4 Content-Length: 0
5 Date: Sat, 30 Sep 2017 12:13:37 GMT
6 X-Powered-By: Express
1  $ http POST http://localhost:4000/v1/users id:=3
2 HTTP/1.1 422 Unprocessable Entity
3 Connection: keep-alive
4 Content-Length: 102
5 Content-Type: application/json; charset=utf-8
6 Date: Sat, 30 Sep 2017 12:13:50 GMT
7 ETag: W/"66-/c4cfoquQZGwtDBUzHjJydJAHJ0"
8 X-Powered-By: Express
10 {
       "error": "Error at array index 0: (ErrorAtProperty \"name\" (TypeMismatch \"String\"
   \"Undefined\"))"
12 }
1  $ http POST http://localhost:4000/v1/users id:=3 name=""
2 HTTP/1.1 422 Unprocessable Entity
3 Connection: keep-alive
4 Content-Length: 39
5 Content-Type: application/json; charset=utf-8
6 Date: Sat, 30 Sep 2017 12:14:02 GMT
7 ETag: W/"27-JQsh12xu/rEFdWy8REF4NMtBUB4"
8 X-Powered-By: Express
9
10 {
      "error": "User name must not be empty"
11
12 }
1  $ http POST http://localhost:4000/v1/users id:=0 name=roger
2 HTTP/1.1 422 Unprocessable Entity
3 Connection: keep-alive
4 Content-Length: 36
5 Content-Type: application/json; charset=utf-8
6 Date: Sat, 30 Sep 2017 12:14:14 GMT
7 ETag: W/"24-Pvt1L4eGilBmVta0GHlSReJ413E"
8 X-Powered-By: Express
10 {
      "error": "User ID must be positive"
11
12 }
```

```
1  $ http GET http://localhost:4000/v1/user/3
2 HTTP/1.1 200 OK
3 Connection: keep-alive
4 Content-Length: 23
5 Content-Type: application/json; charset=utf-8
6 Date: Sat, 30 Sep 2017 12:14:28 GMT
7 ETag: W/"17-1scpiB1FT9DBu9s4I1gNWSjH2go"
  X-Powered-By: Express
10 {
       "id": 3,
11
       "name": "roger"
13 }
1  $ http GET http://localhost:4000/v1/user/asdf
2 HTTP/1.1 422 Unprocessable Entity
3 Connection: keep-alive
4 Content-Length: 38
5 Content-Type: application/json; charset=utf-8
6 Date: Sat, 30 Sep 2017 12:14:40 GMT
7 ETag: W/"26-//tvORl1gGDUMwgSaqbEpJhuadI"
8 X-Powered-By: Express
10 {
       "error": "User ID must be an integer"
11
12 }
1  $ http GET http://localhost:4000/v1/user/-1
2 HTTP/1.1 422 Unprocessable Entity
3 Connection: keep-alive
4 Content-Length: 36
5 Content-Type: application/json; charset=utf-8
6 Date: Sat, 30 Sep 2017 12:14:45 GMT
  ETag: W/"24-Pvt1L4eGilBmVtaOGHlSReJ413E"
  X-Powered-By: Express
10 {
       "error": "User ID must be positive"
11
12 }
```

It works as expected.

## Configuration

Right now our application configuration resides in the main function:

```
main = runServer port databaseConfig
     where
2
       port = 4000
3
       databaseConfig = { user: "abhinav"
                        , password: ""
5
                        , host: "localhost"
6
                         , port: 5432
                         , database: "simple_service"
8
                        , max: 10
                         , idleTimeoutMillis: 1000
10
                        }
11
```

We are going to extract it out of the code and read it from the environment variables using the purescript-config package. First, we install the required packages using bower.

```
1 $ bower install --save purescript-node-process purescript-config
```

Now, we write the following code in the src/SimpleService/Config.purs file:

```
module SimpleService.Config where
2
  import Data.Config
3
  import Prelude
  import Control.Monad.Eff (Eff)
  import Data.Config.Node (fromEnv)
  import Data.Either (Either)
  import Data.Set (Set)
10 import Database.PostgreSQL as PG
  import Node.Process (PROCESS)
11
12
13 type ServerConfig =
    { port
14
                   :: Int
     , databaseConfig :: PG.PoolConfiguration
    }
16
17
18 databaseConfig :: Config {name :: String} PG.PoolConfiguration
19 databaseConfig =
     { user: _, password: _, host: _, port: _, database: _, max: _, idleTimeoutMillis: _ }
    <$> string {name: "user"}
21
    <*> string {name: "password"}
22
    <*> string {name: "host"}
    24
    <*> string {name: "database"}
25
    <*> int {name: "pool_size"}
26
    27
28
29 portConfig :: Config {name :: String} Int
  portConfig = int {name: "port"}
31
32 serverConfig :: Config {name :: String} ServerConfig
33 serverConfig =
    { port: _, databaseConfig: _}
34
    <$> portConfig
35
    <*> prefix {name: "db"} databaseConfig
37
38 readServerConfig :: forall eff.
                     Eff (process :: PROCESS | eff) (Either (Set String) ServerConfig)
40 readServerConfig = fromEnv "SS" serverConfig
```

We use the applicative DSL provided in Data.Config module to build a description of our configuration. This description contains the keys and types of the configuration, for consumption by various interpreters. Then we use the fromEnv interpreter to read the config from the environment variables derived from the name fields in the records in the description in the readServerConfig function. We also write a bash script to set those environment variables in the development environment in the setenv.sh file:

```
1 export SS_PORT=4000
2 export SS_DB_USER="abhinav"
3 export SS_DB_PASSWORD=""
4 export SS_DB_HOST="localhost"
5 export SS_DB_PORT=5432
6 export SS_DB_DATABASE="simple_service"
7 export SS_DB_POOL_SIZE=10
8 export SS_DB_IDLE_CONN_TIMEOUT_MILLIS=1000
```

Now we rewrite our src/Main.purs file to use the readServerConfig function:

```
module Main where
  import Prelude
3
  import Control.Monad.Eff (Eff)
5
   import Control.Monad.Eff.Console (CONSOLE, log)
   import Data.Either (Either(..))
  import Data.Set (toUnfoldable)
   import Data.String (joinWith)
10 import Database.PostgreSQL as PG
import Node.Express.Types (EXPRESS)
import Node.Process (PROCESS)
13 import Node.Process as Process
14 import SimpleService.Config (readServerConfig)
import SimpleService.Server (runServer)
16
   main :: forall eff. Eff ( console :: CONSOLE
17
                           , express :: EXPRESS
18
                           , postgreSQL :: PG.POSTGRESQL
19
                           , process :: PROCESS
                           | eff ) Unit
21
22 main = readServerConfig >>= case _ of
     Left missingKeys -> do
23
       log $ "Unable to start. Missing Env keys: " <> joinWith ", " (toUnfoldable
24
   missingKeys)
       Process.exit 1
25
     Right { port, databaseConfig } -> runServer port databaseConfig
```

If readServerConfig fails, we print the missing keys to the console and exit the process. Else we run the server with the read config.

To test this, we stop the server we ran in the beginning, source the config, and run it again:

```
1  $ pulp --watch run
2  * Building project in /Users/abhinav/ps-simple-rest-service
3  * Build successful.
4  Server listening on :4000
5  ^C
6  $ source setenv.sh
7  $ pulp --watch run
8  * Building project in /Users/abhinav/ps-simple-rest-service
9  * Build successful.
10  Server listening on :4000
```

It works! We test the failure case by opening another terminal which does not have the environment variables set:

Up next, we add logging to our application.

## Logging

For logging, we use the purescript-logging package. We write a logger which logs to stdout; in the src/SimpleService/Logger.purs file:

```
module SimpleService.Logger
     ( debug
     , info
3
     , warn
4
     , error
5
6
     ) where
   import Prelude
8
10 import Control.Logger as L
import Control.Monad.Eff.Class (class MonadEff, liftEff)
import Control.Monad.Eff.Console as C
13 import Control.Monad.Eff.Now (NOW, now)
14 import Data.DateTime.Instant (toDateTime)
15 import Data.Either (fromRight)
import Data.Formatter.DateTime (Formatter, format, parseFormatString)
import Data.Generic.Rep (class Generic)
18 import Data.Generic.Rep.Show (genericShow)
import Data.String (toUpper)
   import Partial.Unsafe (unsafePartial)
21
   data Level = Debug | Info | Warn | Error
22
23
   derive instance eqLevel :: Eq Level
24
   derive instance ordLevel :: Ord Level
   derive instance genericLevel :: Generic Level _
26
27
   instance showLevel :: Show Level where
28
     show = toUpper <<< genericShow</pre>
29
30
31 type Entry =
     { level :: Level
32
     , message :: String
33
34
     }
35
  dtFormatter :: Formatter
   dtFormatter = unsafePartial $ fromRight $ parseFormatString "YYYY-MM-DD HH:mm:ss.SSS"
37
38
   logger :: forall m e. (
39
             MonadEff (console :: C.CONSOLE, now :: NOW | e) m) => L.Logger m Entry
40
   logger = L.Logger $ \{ level, message } -> liftEff do
41
     time <- toDateTime <$> now
42
     \text{C.log} \ "[" <> format dtFormatter time <> "] " <> show level <> " " <> message
43
  log :: forall m e.
45
           MonadEff (console :: C.CONSOLE , now :: NOW | e) m
46
        => Entry -> m Unit
47
   log entry@{level} = L.log (L.cfilter (\e -> e.level == level) logger) entry
48
49
50 debug :: forall m e.
```

```
MonadEff (console :: C.CONSOLE , now :: NOW | e) m => String -> m Unit
51
   debug message = log { level: Debug, message }
53
  info :: forall m e.
           MonadEff (console :: C.CONSOLE , now :: NOW | e) m => String -> m Unit
55
  info message = log { level: Info, message }
56
57
   warn :: forall m e.
58
           MonadEff (console :: C.CONSOLE , now :: NOW | e) m => String -> m Unit
59
   warn message = log { level: Warn, message }
60
61
62 error :: forall m e.
            MonadEff (console :: C.CONSOLE , now :: NOW | e) m => String -> m Unit
63
64 error message = log { level: Error, message }
```

purescript-logging lets us define our own logging levels and loggers. We define four log levels, and a log entry type with the log level and the message. Then we write the logger which will print the log entry to stdout along with the current time as a well formatted string. We define convenience functions for each log level.

Before we proceed, let's install the required dependencies.

```
1 $ bower install --save purescript-logging purescript-now purescript-formatters
```

Now we add a request logger middleware to our server in the src/SimpleService/Server.purs file:

```
-- previous code
   import Control.Monad.Eff.Console (CONSOLE)
  import Control.Monad.Eff.Now (NOW)
  import Data.Maybe (maybe)
   import Data.String (toUpper)
   import Node.Express.App (App, all, delete, get, http, listenHttp, post, use,
   useExternal, useOnError)
   import Node.Express.Handler (Handler, next)
   import Node.Express.Request (getMethod, getPath)
   import SimpleService.Logger as Log
  -- previous code
10
  requestLogger :: forall eff. Handler (console :: CONSOLE, now :: NOW | eff)
12
13 requestLogger = do
     method <- getMethod
     path <- getPath
15
16
     Log.debug $ "HTTP: " <> maybe "" id ((toUpper <<< show) <$> method) <> " " <> path
17
18
19 app :: forall eff.
          PG.Pool
20
       -> App (postgreSQL :: PG.POSTGRESQL, console :: CONSOLE, now :: NOW | eff)
21
22 app pool = do
     useExternal jsonBodyParser
23
     use requestLogger
24
     -- previous code
```

We also convert all our previous logging statements which used Console.log to use SimpleService.Logger and add logs in our handlers. We can see logging in effect by restarting the server and hitting it:

```
1  $ pulp --watch run
2  * Building project in /Users/abhinav/ps-simple-rest-service
3  * Build successful.
4  [2017-09-30 16:02:41.634] INFO Server listening on :4000
5  [2017-09-30 16:02:43.494] DEBUG HTTP: PATCH /v1/user/3
6  [2017-09-30 16:02:43.517] DEBUG Updated user: 3
7  [2017-09-30 16:03:46.615] DEBUG HTTP: DELETE /v1/user/3
8  [2017-09-30 16:03:46.635] DEBUG Deleted user 3
9  [2017-09-30 16:05:03.805] DEBUG HTTP: GET /v1/users
```

#### Conclusion

In this tutorial we learned how to create a simple JSON REST web service written in PureScript with persistence, validation, configuration and logging. The complete code for this tutorial can be found in github. Discuss this post in the comments.